Solmu 3/2001



Frégier'n lause

Simo K. Kivelä

Kevään 2001 ylioppilaskirjoitusten pitkän matematiikan kokeessa oli seuraava tehtävä:

> Suorakulmaisen kolmion kaikki kärjet sijaitsevat paraabelilla $y = x^2$; suoran kulman kärki on paraabelin huipussa. Osoita, että jokaisen tällaisen kolmion hypotenuusa leikkaa paraabelin akselin samassa pisteessä. Määritä tämä piste.

Kyseessä on yleisemminkin voimassa oleva tulos, jota kutsutaan *Frégier'n lauseeksi*. Suoran kulman kärjen ei tarvitse olla paraabelin huipussa eikä käyränkään tarvitse olla paraabeli. Se voi olla mikä tahansa toisen asteen käyrä – ellipsi, paraabeli tai hyperbeli – ja suoran kulman kärki voi olla mikä tahansa sen piste. Tällöin kolmion hypotenuusa kulkee aina saman pisteen kautta. Paraabelin akselia vastaa tällöin sellainen käyrän normaali, joka kulkee suoran kulman kärkipisteen kautta. Yleinen Frégier'n lause kuuluu siten seuraavasti:

> Olkoon P toisen asteen käyrällä sijaitseva piste ja olkoot PA ja PB kaksi tämän pisteen kautta kulkevaa toisiaan vastaan kohtisuoraa käyrän jännettä. Tällöin on olemassa kiinteä piste Q, jonka kautta suora AB kulkee riippumatta siitä, missä asennossa jänteet PA ja PB ovat. Piste Q sijaitsee pisteeseen P asetetulla käyrän normaalilla.

Kuva tilanteesta ellipsitapauksessa on edempänä.

Frégier'n lauseen todistus voi perustua analyyttiseen geometriaan. Ylioppilastehtävässä tämä on hyvinkin yksinkertainen, mutta esimerkiksi ellipsitapauksessa laskuista tulee jonkin verran hankalia. Tämän johdosta on luontevaa käyttää hyväksi jotakin symbolisen laskennan ohjelmaa, jolloin ei tarvitse huolehtia monimutkaisista sievennyksistä ja näissä herkästi syntyvistä virheistä. Ei symbolisten ohjelmienkaan käyttö ongelmatonta ole: niillä on omat heikkoutensa, joita on opittava varomaan.

Seuraavassa esitetään Frégier'n lauseen todistus ellipsitapauksessa Mathematica-nimistä symbolista ohjelmaa käyttäen, jolloin mekaaniset laskut voidaan antaa ohjelman tehtäviksi. Itse asiassa tämä kirjoitus on kokonaisuudessaan laadittu Mathematican avulla; se nimittäin sisältää myös ainakin jonkintasoiset mahdollisuudet tekstinkäsittelyyn ja matemaattisten kaavojen kirjoittamiseen.

Todistus

Tarkasteltavana oleva ellipsi voidaan sijoittaa origokeskiseksi, ilman että probeemaa mitenkään rajoitetaan. Muodostetaan siis aluksi origokeskisen ellipsin yhtälö ja talletetaan tämä nimelle `ellipsi`:

```
ellipsi = x^2 / a^2 + y^2 / b^2 == 1
```

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} == 1$$

Lauseessa esiintyvä piste P valitaan ellipsin kehältä. Jokainen kehäpiste voidaan esittää muodossa $(a\cos t, b\sin t)$. Tässä t on ns. parametri, joka voi saada arvot väliltä $0 \le t < 2\pi$; tällöin jokainen kehäpiste tulee huomioiduksi. Annetaan pisteen koordinaateille lyhyemmät nimet `x0` ja `y0`, jolloin niihin on helpompi myöhemmin viitata.

```
{x0, y0} = {a Cos[t], b Sin[t]}

{a Cos[t], b Sin[t]}
```

Pisteen koordinaatit voidaan sijoittaa ellipsin yhtälöön, jolloin se todellakin toteutuu:

```
ellipsi /. {x -> x0, y -> y0} // Simplify

True
```

Tämän pisteen kautta asetetaan kaksi toisiaan vastaan kohtisuoraa suoraa. Annetaan näiden yhtälöt vektorimuodossa, jolloin niitä on näppärää käsitellä. Vektorit esitetään Mathematicassa kahden alkion listoina, esimerkiksi $2\bar{\imath} + 3\bar{\jmath}$ muodossa `{2,3}`. Koska suorat ovat kohtisuoria, tulee niiden suuntavektoreiden skalaaritulon olla $= 0$. Jos suuntavektorit kirjoitetaan muotoon `{p,q}` ja `{-q,p}` tämä ehto täyttyy, mutta muulla tavoin ei suorien suuntia ole rajoitettu. Vektoriesityksessä tarvittava parametri olkoon u:

```
suora1 = {x, y} == {x0, y0} + u {p, q}

{x, y} == {p u + a Cos[t], q u + b Sin[t]}

suora2 = {x, y} == {x0, y0} + u {-q, p}

{x, y} == {-q u + a Cos[t], p u + b Sin[t]}
```

Etsitään erikseen kummankin suoran ja ellipsin leikkauspisteet. Näitä on kummassakin tapauksessa kaksi ja niistä toinen on luonnollisesti ellipsillä oleva piste P eli `{x0,y0}`, jonka kautta suorat asetettiin. Ratkaisujen sieventäminen edellyttää useampaa sievennyskäskyä peräkkäin aseteltuina:
`//FullSimplify//PowerExpand//FullSimplify` :

```
ratk1 = Solve[{ellipsi, suora1}, {x, y, u}] // FullSimplify //
   PowerExpand // FullSimplify
```

$$\Big\{\Big\{u \to -\frac{2\,a\,b\,(b\,p\,\mathrm{Cos}[t] + a\,q\,\mathrm{Sin}[t])}{b^2\,p^2 + a^2\,q^2},$$
$$x \to \frac{a\,((-b^2\,p^2 + a^2\,q^2)\,\mathrm{Cos}[t] - 2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{b^2\,p^2 + a^2\,q^2},$$
$$y \to \frac{b\,(-2\,a\,b\,p\,q\,\mathrm{Cos}[t] + (b\,p - a\,q)\,(b\,p + a\,q)\,\mathrm{Sin}[t])}{b^2\,p^2 + a^2\,q^2}\Big\},$$
$$\{u \to 0,\ x \to a\,\mathrm{Cos}[t],\ y \to b\,\mathrm{Sin}[t]\}\Big\}$$

```
ratk2 = Solve[{ellipsi, suora2}, {x, y, u}] // FullSimplify //
   PowerExpand // FullSimplify
```

$$\left\{\{u\to 0,\ x\to a\,\mathrm{Cos}[t],\ y\to b\,\mathrm{Sin}[t]\},\ \left\{u\to \frac{2\,a\,b\,(b\,q\,\mathrm{Cos}[t]-a\,p\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2},\right.\right.$$
$$x\to \frac{a\,((a\,p-b\,q)\,(a\,p+b\,q)\,\mathrm{Cos}[t]+2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2},$$
$$\left.\left. y\to \frac{b\,(2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(-a^2\,p^2+b^2\,q^2)\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2}\right\}\right\}$$

Edellisessä tapauksessa edellinen ratkaisu antaa etsityn pisteen. Jälkimmäisessä tapauksessa tämä on jälkimmäinen. Talletetaan saatujen toisten leikkauspisteiden A ja B koordinaatit omille nimilleen:

```
{x1, y1} = {x, y} /. First[ratk1]
```

$$\left\{\frac{a\,((-b^2\,p^2+a^2\,q^2)\,\mathrm{Cos}[t]-2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2},\ \frac{b\,(-2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(b\,p-a\,q)\,(b\,p+a\,q)\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2}\right\}$$

```
{x2, y2} = {x, y} /. Last[ratk2]
```

$$\left\{\frac{a\,((a\,p-b\,q)\,(a\,p+b\,q)\,\mathrm{Cos}[t]+2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2},\ \frac{b\,(2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(-a^2\,p^2+b^2\,q^2)\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2}\right\}$$

Saatujen pisteiden kautta kulkeva suora on kolmion hypotenuusan määräämä suora. Muodostetaan sille tavanomainen x- ja y-koordinaatit sitova yhtälö:

```
hypo = y - y1 == (y2 - y1) / (x2 - x1) (x - x1)
```

$$y-\frac{b\,(-2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(b\,p-a\,q)\,(b\,p+a\,q)\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2} ==$$
$$\left(\left(x-\frac{a\,((-b^2\,p^2+a^2\,q^2)\,\mathrm{Cos}[t]-2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2}\right)\right.$$
$$\left(-\frac{b\,(-2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(b\,p-a\,q)\,(b\,p+a\,q)\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2}\right.$$
$$\left.\left.\frac{b\,(2\,a\,b\,p\,q\,\mathrm{Cos}[t]+(-a^2\,p^2+b^2\,q^2)\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2}\right)\right)\Big/$$
$$\left(-\frac{a\,((-b^2\,p^2+a^2\,q^2)\,\mathrm{Cos}[t]-2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{b^2\,p^2+a^2\,q^2}+\right.$$
$$\left.\frac{a\,((a\,p-b\,q)\,(a\,p+b\,q)\,\mathrm{Cos}[t]+2\,a\,b\,p\,q\,\mathrm{Sin}[t])}{a^2\,p^2+b^2\,q^2}\right)$$

Lukija saattaa ihmetellä, miksi toisinaan suoralle käytetään vektorimuotoista, toisinaan xy-muotoista yhtälöä. Molemmat ovat periaatteessa yhtä hyviä. Edellä olevat valinnat on tehty tavoitteena mahdollisimman yksinkertaiset ja toisaalta symbolisen ohjelman käyttömahdollisuuksia mahdollisimman hyvin valaisevat laskut.

On ehkä aika piirtää tilanteesta kuvio. Tätä varten on aluksi ladattava Mathematican lisäpaketti:

```
Needs["Graphics`ImplicitPlot`"]
```

Piirtämistä varten muodostetaan suorille PA ja PB xy-muotoiset yhtälöt eliminoimalla vektoriesityksestä parametri:

```
suora1xy = Eliminate[suora1, u]

p y + a q Cos[t] - b p Sin[t] == q x

suora2xy = Eliminate[suora2, u]

-q y + a p Cos[t] + b q Sin[t] == p x
```

Kuvio piirretään seuraavilla lukuarvoilla:

```
lukuarvot = {a -> 3, b -> 2, t -> 1.2, p -> 2, q -> 3}

{a → 3, b → 2, t → 1.2, p → 2, q → 3}
```

Lähtökohtana oleva piste P on tällöin

```
{x0, y0} /. lukuarvot

{1.08707, 1.86408}
```

Tämän jälkeen voidaan piirtää itse kuva:

```
kuva1 = ImplicitPlot[
  Evaluate[{ellipsi, suora1xy, suora2xy, hypo} /. lukuarvot],
  {x, -5, 5}, {y, -5, 5}]
```



Frégier'n lause väittää, että hypotenuusasuora kulkee aina saman pisteen kautta riippumatta kohtisuorien suorien asennosta. Miten tämä piste voidaan löytää?

Yhtenä mahdollisuutena on toistaa edellä oleva lasku siten, että suorien PA ja PB suunnat ovat erilaiset (mutta edelleen toisiaan vastaan kohtisuorat), jolloin saadaan jokin toinen hypotenuusasuora. Kahden hypotenuusasuoran leikkauspisteestä saadaan ainakin sopiva ehdokas etsityksi pisteeksi.

Uudet suuntavektorit olkoot `{r,s}` ja `{-s,r}`. Uusi hypotenuusasuora saadaan hyvin yksinkertaisesti: korvataan aiemmassa hypotenuusasuorassa vanhat suuntavektorin komponentit uusilla:

```
hypo2 = hypo /. {p -> r, q -> s}
```

$$y - \frac{b\,(-2\,a\,b\,r\,s\,\mathrm{Cos}[t] + (b\,r - a\,s)\,(b\,r + a\,s)\,\mathrm{Sin}[t])}{b^2\,r^2 + a^2\,s^2} == \left(\left(x - \frac{a\,((-b^2\,r^2 + a^2\,s^2)\,\mathrm{Cos}[t] - 2\,a\,b\,r\,s\,\mathrm{Sin}[t])}{b^2\,r^2 + a^2\,s^2}\right)\left(-\frac{b\,(-2\,a\,b\,r\,s\,\mathrm{Cos}[t] + (b\,r - a\,s)\,(b\,r + a\,s)\,\mathrm{Sin}[t])}{b^2\,r^2 + a^2\,s^2} \frac{b\,(2\,a\,b\,r\,s\,\mathrm{Cos}[t] + (-a^2\,r^2 + b^2\,s^2)\,\mathrm{Sin}[t])}{a^2\,r^2 + b^2\,s^2}\right)\right) \Big/ \left(-\frac{a\,((-b^2\,r^2 + a^2\,s^2)\,\mathrm{Cos}[t] - 2\,a\,b\,r\,s\,\mathrm{Sin}[t])}{b^2\,r^2 + a^2\,s^2} + \frac{a\,((a\,r - b\,s)\,(a\,r + b\,s)\,\mathrm{Cos}[t] + 2\,a\,b\,r\,s\,\mathrm{Sin}[t])}{a^2\,r^2 + b^2\,s^2}\right)$$

Tämän jälkeen voidaan ratkaista hypotenuusasuorien leikkauspiste ja sievennetään se:

```
ratk = Solve[{hypo, hypo2}, {x, y}]

piste = {x, y} /. ratk[[1]] // Simplify
```

$$\left\{\frac{a\,(a^2 - b^2)\,\mathrm{Cos}[t]}{a^2 + b^2},\ \frac{b\,(-a^2 + b^2)\,\mathrm{Sin}[t]}{a^2 + b^2}\right\}$$

Sieventämättömänä tulos on monimutkainen[1], mutta se sievenee yllättäen sangen yksinkertaiseen muotoon. Lopputulos näyttää lisäksi olevan riippumaton suuntavektoreiden komponenteista `p`, `q`, `r` ja `s`. **Mutta tämä merkitsee, että Frégier'n lause on tullut todistetuksi:** On löytynyt kahdella hypotenuusasuoralla sijaitseva piste, joka ei riipu lähtökohtana olleista suuntavektoreista; se siis sijaitsee hypotenuusasuoralla valittiinpa suuntavektorit miten tahansa!

Ratkaisu ei luonnollisestikaan mene edellä esitetyllä tavalla, jos `p = r` ja `q = s`. Tätä ei laskussa ole mitenkään suljettu pois. Mathematica kuitenkin käsittelee tällaisessa tilanteessa ns. geneeristä tapausta, ts. tapausta, missä mitään rajoittavia erikoisehtoja ei vallitse. Itse asiassa oletetaan siis `p ~= r`, `q ~= s`.

Lukija miettiköön, mistä seuraa, että löydetty piste on pisteen P kautta kulkevalla ellipsin normaalilla. Solmun verkkoversiossa oleva animaatio antanee ainakin viitteitä.

Esimerkkitapauksessa hypotenuusasuorien leikkauspiste on

```
numpiste = piste /. lukuarvot

{0.418105, -0.716953}
```

[1] Tämän voi todeta Solmun verkkoversiosta.

Tilanteesta saadaan havainnollisempi kuva piirtämällä kaksi keskenään kohtisuoraa suoraparia, näitä vastaavat hypotenuusasuorat ja hypotenuusasuorien leikkauspiste:

```
lukuarvot2 = {a -> 3, b -> 2, t -> 1.2, p -> 3, q -> 1}

{a → 3, b → 2, t → 1.2, p → 3, q → 1}

kuva2 = ImplicitPlot[
   Evaluate[{ellipsi, suora1xy, suora2xy, hypo} /. lukuarvot2],
   {x, -5, 5}, {y, -5, 5}, DisplayFunction → Identity];
Show[kuva1, kuva2, Graphics[{PointSize[0.03], Point[numpiste]}]]
```



Solmun verkkoversiossa asiaa havainnollistetaan myös animaatiolla.